
7.1 3D図形の描画
3D基本設定
p size(幅, 高さ, P3D)

n ウィンドウを3D用で開く
p lights()

n 標準の照明を設定
n draw()のなかで最初に書く

p perspective()
n 透視投影に設定（第9回）

3D基本形状
p box(辺の長さ)
p box(幅, 高さ, 奥行き)

n 原点に立方体/直方体を描画
p sphere(半径)

n 原点に球を描画
n 通常は noStroke() で描く

3次元座標系（無指定時）
p Processingではz軸は手前方向

7.2* 座標系のとり方（p.32）
p 左手系

n 視点座標系・CGゲーム
n DirectX

p 左手系
n Processing

p 右手系
n CG理論・数学・工学分野
n OpenGL

p 右手系
n 建築座標系

7.3 3Dでの位置設定
3Dでの位置設定
p 座標変換を駆使せよ

n 3DCGでは，幾何変換で図形を
配置する考え方が必須!!

n boxもsphereもそのときの描画
座標系の原点付近に図形を描く

n 「原点」と「拡大率」を常に意識！

行列スタックの操作
p pushMatrix() 

n システム変換行列（論理座標
系）を一時的に退避する

n 使い方は，2次元と同じ
p popMatrix()

n 最近保存した論理座標系を戻す
n pushとpopは必ず対にすること

3次元幾何変換
p translate(tx, ty, tz)

n 座標系の平行移動
n 最初に (width/2, height/2, 0)

に原点をもってくると分かりやすい
p scale(sx, sy, sz)

n 座標系の拡大・縮小
n 原点を中心に全体が拡大

p rotateX(θx)
n x軸まわりの回転

p rotateY(θy)
n y軸まわりの回転

p rotateZ(θz)
n z軸まわりの回転
n 2次元のrotate(θz) と同じ

7.4 3D描画の例

void setup() {
// P3Dモードでウィンドウを開く
size(400, 400, P3D);
noLoop();

}

void draw() {
background(0);
// 標準の照明
lights();
// 透視投影
perspective();
// 原点を移動
translate(width/2, height/2, 0);
noStroke();
fill(255, 200, 200);
// 原点に半径100の球を描画
sphere(100);

}

// 回転する立方体
float rot = 0.0;

void setup() {
size(400, 400, P3D);

}

void draw() {
background(70);
lights();
perspective();
translate(width/2, height/2, 0);
pushMatrix();

rotateY(radians(rot++));
stroke(255, 0, 0);
fill(255, 255, 0);
box(100);

popMatrix();
}



7.5* モデリングの基礎
モデリング
p モデリングとは（p.33）

n 3Dオブジェクト（物体）の形状を
数値データの集合で表すこと

形状モデル（p.60）
p ワイヤーフレームモデル

n 線の集合で物体を表現する
p サーフェスモデル

n 物体の表面（だけ）を表す
n 通常はポリゴン（多角形）の集合

p ソリッドモデル
n 物体の内外を示す情報もあり，

中身が詰まっているモデル
p ポイントクラウド（点群）

n 点（＋色）の集合によるデータ

簡単なモデリング
p ポリゴンの描画

n ポリゴンpolygon ＝ 多角形
n 物体表面のポリゴンを描画する

（beginShape～endShape）
p 例）三角柱

n 幅＝2（-1≦x≦1），原点に重心，
高さ＝2（-1≦y≦1）

7.6 ポリゴンの描画例
// 三角柱を表示する
void setup() {
size(400, 400, P3D);

}

void draw() {
background(0);
lights();
perspective();
translate(width/2, height/2);
pushMatrix();
rotateX(radians(frameCount)/2);
rotateY(radians(frameCount));
// noFill()ならワイヤーフレーム表示
fill(255, 255, 0);
stroke(100, 255, 100);
// 底面の幅と高さを指定して描画
prism3(60, 120);

popMatrix();
}

void prism3(float w, float h) {
w /= 2; h /= 2;
float g = sqrt(3.0) / 3.0 * w;
// 側面の3枚の長方形
beginShape(QUADS);
vertex(w, -h, -g); vertex(w, h, -g);
vertex(0, h, g*2); vertex(0, -h, g*2);
vertex(0, -h, g*2); vertex(0, h, g*2);
vertex(-w, h, -g); vertex(-w, -h, -g);
vertex(-w, -h, -g); vertex(-w, h, -g);
vertex(w, h, -g); vertex(w, -h, -g);

endShape();
// 底面と上面の三角形
beginShape(TRIANGLES);
vertex(w, -h, -g); vertex(0, -h, g*2);
vertex(-w, -h, -g);
vertex(w, h, -g); vertex(0, h, g*2);
vertex(-w, h, -g);

endShape();
}

7.7 3Dモデルデータの利用
3Dモデル表示
p PShape型

n P3DではOBJデータが利用可能
（2DのPShapeは第3回資料参照）

p モデルの読み込み
n m = loadShape("ファイル名")
n 2D同様，setup()内で準備する

p モデルの表示
n shape(m)
n shape(m, x, y, z)

p モデルの幾何変換
n モデルデータ内における変換
n m.translate(…)
n m.scale(…)
n m.rotate{X,Y,Z}(…)

PShapeのメソッド
p 描画色の設定

n m.setFill(色or真偽値)
n m.setStroke(色or真偽値)
n 例） noFill() に相当することは，

m.setFill(false) とする
p 入れ子モデルの取得

n モデルの中に階層的にモデルが
入っていることがある

n int n = m.getChildCount()
n PShape c = m.getChild(i)

p 頂点座標の取得
n その階層の頂点を順に取得
n int n = m.getVertexCount()
n PVector v = m.getVertex(i)
n PVector型はx,y,z座標を持つ

7.8 3Dモデルデータの表示例
// 準備: beethoven.zip をダウンロードし，
// 中身の3ファイルをdataフォルダに入れる
PShape model;

void setup() {
size(400, 400, P3D);
model = loadShape("beethoven.obj");
// テクスチャを無効化にして色をつける例
//model.setTexture(null);
//model.setFill(true);
//model.setFill(color(255, 0, 0));

}

void draw() {
background(0, 0, 128);
lights();
translate(width/2, height/2, 0);
rotateX(PI);
rotateY(radians(frameCount));

scale(200);
// 通常のモデル表示
shape(model);
// 頂点のみを表示する例
//stroke(255); strokeWeight(0.01);
//shapeAsPoints(model);

}

// 頂点情報を取り出して点群で描画する例
void shapeAsPoints(PShape sh) {
int n = sh.getVertexCount();
for (int i = 0; i < n; i++) {
PVector v = sh.getVertex(i);
point(v.x, v.y, v.z);

}
n = sh.getChildCount();
for (int i = 0; i < n; i++)
shapeAsPoints(sh.getChild(i));

}

7.9 演習課題
課題
問1) 正八面体を描画するプログラム

を作成しなさい
n 8枚の正三角形を描画する

（beginShapeでTRIANGLES）
n 回転させて裏側も確認しよう
n 他の図形を“追加”してもよい

問2) 《前回の復習》
2次元幾何変換A～Cについて以下の
問いに答え，A4用紙で提出しなさい

1. 合成変換行列ABを計算しなさい
2. 変換ABの後に座標 (20, 60) に点を

打つと，画面のどこに表示されるか？
3. 合成変換行列BAを計算し，ABとの意

味の違いを説明しなさい
4. 行列Cに対応するProcessingの命令

を示しなさい（定数PIを用いてもよい）
5. 合成変換行列C2=CCを計算し，どの

ような変換か説明しなさい

7.10 参考：点群による球の描画
PImage img;  // 球の表面画像

void setup() {
size(600, 600, P3D);
img = loadImage("earth.jpg");
img.loadPixels();

}

void draw() {
background(0);
lights(); perspective();
translate(width/2, height/2);
pushMatrix();
rotateX(radians(frameCount)/2);
rotateY(radians(frameCount));
pointSphere(200, 2, img);

popMatrix();
}

// 点群による球面の描画例
void pointSphere(float r, int d, PImage g) {
strokeWeight(d * 4);

// 緯度（lat）と経度（lng）による2重ループ
for (int lat = 90 - d; lat > -90; lat -= d) {
float a = radians(lat);
int e = (int)(d / cos(a));  // 緯度で間隔調整
for (int lng = -180; lng < 180; lng += e) {

// 表面画像の対応点から色を抽出
int u = (lng + 180) * g.width / 360;
int v = (-lat + 90) * g.height / 180;
stroke(g.pixels[u + v * g.width]);

// 緯度・経度から球面の3D座標を計算
float b = radians(lng);
point(r * cos(a) * cos(b), r * sin(a),

r * cos(a) * sin(b));
}

}
}

7.11 参考：3DCGソフトウェア紹介
p MagicaVoxel ←おすすめ

n ephtracy.github.io
n Minecraftのようにボクセル

（立方体）でモデリング
p Tinkercad ←おすすめ

n www.tinkercad.com
n インストール不要なWebアプリ

p SketchUp Free
n www.sketchup.com
n 建物・人工物のモデリングに向く

p ScupltGL
n stephaneginier.com/sculptgl/
n 粘土・彫刻のようにモデリング

p Blender
n www.blender.org
n 高機能でフリー&オープンソース

p Maya / 3ds Max など
n Autodesk社のプロ向け製品
n 学生は無償で利用可能
n www.autodesk.co.jp/education

p メタセコイア
n www.metaseq.net
n 日本製で資料が豊富

p Sculptris
n pixologic.com/sculptris/
n 粘土・彫刻のようにモデリング

p Vue Pioneer
n www.e-onsoftware.com
n 自然景観生成（非商用フリー版）

p DAZStudio
n www.daz3d.com/get_studio
n 人体ポーズ&アニメーション作成

7.11 参考：3Dモデルデータの取得
p Google Poly

n poly.google.com
p Free3D

n free3d.com
p 3DModelFree.com

n www.3dmodelfree.com

p Unityアセットストア
n assetstore.unity.com/search?q=3d&q=models&q=price:0
n OBJ出力
assetstore.unity.com/packages/tools/utilities/scene-obj-exporter-22250

p SketchUp 3D Warehouse
n 3dwarehouse.sketchup.com
n OBJ出力できるプラグイン
sketchup-onigiri.jimdo.com/sketchup-plugins/su2objmtl/

p TurboSquid
n www.turbosquid.com

p Artist-3D.com
n artist-3d.com

p XOIO Air
n xoio-air.de

